CORRECTION (PARTIELLE) DES TP 3 ET 4

1) Mise en évidence de la sensibilité de la méthode Newton à la C.I. (TP3, Ex1):

x0=[];x20=[];n=2000;

for i=1:n+1

   x=(i-1)/n;x0=[x0;x];

   for j=1:20

      x=2*x^3/(3*x^2-1);

   end

   x20=[x20;x];

end

xbasc()

plot2d(x0,x20,1,'011',' ',[0,-2,1,2])

xtitle('sensibilite de la methode de Newton aux...

conditions initiales','x0','x20');

2) Mise en évidence de la rapidité de convergence de la méthode Newton (TP3, Ex1):
x0=40;err=[abs(x0-1)];n=0;

while (x0~=1)

  n=n+1;

  x0=2*x0^3/(3*x0^2-1);

  err=[err;abs(x0-1)];

end

xbasc();

plot2d1('gnl',(0:1:n-1)',err(1:n));

xtitle('erreur de la methode de Newton','n','erreur');

//

// autre possibilite: plot2d((0:1:n-1)',log(err(1:n)))

//
3) Résolution de l’EDO de Bessel d’ordre 0  (TP3, Ex3) :
dt=evstr(x_dialog('pas de temps','0.1'));

deff('ydot=f(t,y)','ydot=[y(2);-y(1)-y(2)/(t+1E-10)]');

t0=0;y0=[1;0];tmax=10;t=t0:dt:tmax;

// methode de resolution avec ode

y=ode(y0,t0,t,f);

// methode de resolution avec Euler explicite

y2=y0;

for i=1:int(tmax/dt)

  y0=y0+dt*f((i-1)*dt,y0);y2=[y2,y0];

end

xbasc()

plot2d([t',t',t'],[y(1,:)',y2(1,:)',besselj(0,t)],[1,2,3],...

'121','solution ode@solution Euler@bessel0');

4) Compraison des méthodes de résolution de Gauss avec pivôt et LU (TP4, Ex1):
nmax=15;err=[];errLU=[];

for n=1:nmax

   A=[10^(-n),1;1,1];b=[1;2];

   L=[1,0;10^n,1];U=[10^(-n),1;0,1-10^n];

   uexact=[-1/(10^(-n)-1);(2*10^(-n)-1)/(10^(-n)-1)]; 

   u1=linsolve(A,-b);

   w=linsolve(L,-b);u=linsolve(U,-w);

   err=[err;max(abs(u1-uexact))];

   errLU=[errLU;max(abs(u-uexact))];

end

xbasc()

plot2d1('gnl',[1:nmax;1:nmax]',[err,errLU])

xtitle('erreur commise par la methode de gauss...

 avec pivot (noir)ou LU (bleu)','n','erreur');

5) Comparaison des vitesses de convergence des méthodes de Jacobi, Gauss Seidel et SOR (TP4, Ex2):
n=evstr(x_dialog('taille de la matrice','50'));

eps=evstr(x_dialog('precision souhaitee','1E-10'));

A=toeplitz([4,1,zeros(1,n-2)]);

D=diag(diag(A));E=-triu(A)+D;F=-tril(A)+D;

uexact=ones(n,1);uinit=rand(n,1);errinit=max(abs(uinit-uexact));

b=A*uexact;

//Jacobi

M1=inv(D);N1=D-A;

u=uinit;errj=errinit;

while(errj($)>eps)

u=M1*(N1*u+b);

errj=[errj;max(abs(u-uexact))];

end

xbasc();

plot2d1('gnl',(1:prod(size(errj)))',errj)

//Gauss Seidel

M2=inv(D-E);N2=F;

u=uinit;errgs=errinit;

while(errgs($)>eps)

u=M2*(N2*u+b);

errgs=[errgs;max(abs(u-uexact))];

end

plot2d1('gnl',(1:prod(size(errgs)))',errgs,2,'001')

//relaxation optimale

rhoj=max(abs(spec(M1*N1)));

omeg=2/(1+sqrt(1-rhoj^2));

M3=inv(1/omeg*D-E);

N3=(1-omeg)/omeg*D+F;

u=uinit;errrel=errinit;

while(errrel($)>eps)

u=M3*(N3*u+b);

errrel=[errrel;max(abs(u-uexact))];

end

plot2d1('gnl',(1:prod(size(errrel)))',errrel,3,'001')

xtitle('comparaison des vitesses de convergence Jacobi...

(noire), gauss Seidel(bleue), relaxation(verte)','n','erreur');

